NAG Library Routine Document

F16EHF (BLAS_DWAXPBY)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F16EHF (BLAS_DWAXPBY) computes the sum of two scaled vectors, preserving input, for real scalars and vectors.

2 Specification

```
SUBROUTINE F16EHF (N, ALPHA, X, INCX, BETA, Y, INCY, W, INCW)
INTEGER            N, INCX, INCY, INCW
REAL (KIND=nag_wp) ALPHA, X(1+(N-1)*ABS(INCX)), BETA,                    &
                   Y(1+(N-1)*ABS(INCY)), W(1+(N-1)*ABS(INCW))
```

The routine may be called by its BLAST name ***blas_dwaxpby***.

3 Description

F16EHF (BLAS_DWAXPBY) performs the operation

$$w \leftarrow \alpha x + \beta y,$$

where x and y are n-element real vectors, and α and β are real scalars.

4 References

Basic Linear Algebra Subprograms Technical (BLAST) Forum (2001) *Basic Linear Algebra Subprograms Technical (BLAST) Forum Standard* University of Tennessee, Knoxville, Tennessee http://www.netlib.org/blas/blast-forum/blas-report.pdf

5 Parameters

1: N – INTEGER *Input*

On entry: n, the number of elements in x, y and w.

2: ALPHA – REAL (KIND=nag_wp) *Input*

On entry: the scalar α.

3: X$(1 + (\text{N} - 1) \times |\text{INCX}|)$ – REAL (KIND=nag_wp) array *Input*

On entry: the n-element vector x.

If INCX > 0, x_i must be stored in X$((i-1) \times |\text{INCX}| + 1)$, for $i = 1, 2, \ldots, \text{N}$.

If INCX < 0, x_i must be stored in X$((\text{N} - i) \times |\text{INCX}| - 1)$, for $i = 1, 2, \ldots, \text{N}$.

Intermediate elements of X are not referenced.

4: INCX – INTEGER *Input*

On entry: the increment in the subscripts of X between successive elements of x.

Constraint: INCX $\neq 0$.

5: BETA – REAL (KIND=nag_wp) *Input*

On entry: the scalar β.

6: Y$(1 + (\mathrm{N} - 1) \times |\mathrm{INCY}|)$ – REAL (KIND=nag_wp) array *Input*

On entry: the n-element vector y.

If INCY > 0, y_i must be stored in Y$(1 + (i - 1) \times \mathrm{INCY})$, for $i = 1, 2, \ldots, \mathrm{N}$.

If INCY < 0, y_i must be stored in Y$(1 - (\mathrm{N} - i) \times \mathrm{INCY})$, for $i = 1, 2, \ldots, \mathrm{N}$.

Intermediate elements of Y are not referenced.

7: INCY – INTEGER *Input*

On entry: the increment in the subscripts of Y between successive elements of y.

Constraint: INCY $\neq 0$.

8: W$(1 + (\mathrm{N} - 1) \times |\mathrm{INCW}|)$ – REAL (KIND=nag_wp) array *Output*

On exit: the n-element vector w.

If INCW > 0, w_i is in W$(1 + (i - 1) \times \mathrm{INCW})$, for $i = 1, 2, \ldots, \mathrm{N}$.

If INCW < 0, w_i is in W$(1 + (\mathrm{N} - i) \times \mathrm{INCW})$, for $i = 1, 2, \ldots, \mathrm{N}$.

Intermediate elements of W are not referenced.

9: INCW – INTEGER *Input*

On entry: the increment in the subscripts of W between successive elements of w.

Constraint: INCW $\neq 0$.

6 Error Indicators and Warnings

If INCX $= 0$ or INCY $= 0$ or INCW $= 0$, an error message is printed and program execution is terminated.

7 Accuracy

The BLAS standard requires accurate implementations which avoid unnecessary over/underflow (see Section 2.7 of Basic Linear Algebra Subprograms Technical (BLAST) Forum (2001)).

8 Parallelism and Performance

Not applicable.

9 Further Comments

None.

10 Example

This example computes the result of a scaled vector accumulation for

$$\begin{aligned} \alpha &= 3, & x &= (-4, 2.1, 3.7, 4.5, -6)^{\mathrm{T}}, \\ \beta &= -1, & y &= (-3, -2.4, 6.4, -5, -5.1)^{\mathrm{T}}. \end{aligned}$$

10.1 Program Text

```
    Program f16ehfe

!     F16EHF Example Program Text

!     Mark 25 Release. NAG Copyright 2014.

!     .. Use Statements ..
      Use nag_library, Only: blas_dwaxpby, nag_wp
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter              :: nin = 5, nout = 6
!     .. Local Scalars ..
      Real (Kind=nag_wp)              :: alpha, beta
      Integer                         :: incw, incx, incy, n, nw, nx, ny
!     .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable  :: w(:), x(:), y(:)
!     .. Intrinsic Procedures ..
      Intrinsic                       :: abs
!     .. Executable Statements ..
      Write (nout,*) 'F16EHF Example Program Results'

!     Skip heading in data file
      Read (nin,*)

      Read (nin,*) n
      Read (nin,*) incx, incy, incw

      nw = 1 + (n-1)*abs(incw)
      nx = 1 + (n-1)*abs(incx)
      ny = 1 + (n-1)*abs(incy)
      Allocate (w(nw),x(nx),y(ny))

      Read (nin,*) alpha, beta
      Read (nin,*) x(1:nx:abs(incx))
      Read (nin,*) y(1:ny:abs(incy))

!     Compute W = alpha*X + beta*Y

      Call blas_dwaxpby(n,alpha,x,incx,beta,y,incy,w,incw)

      Write (nout,*)
      Write (nout,99999)
      Write (nout,99998) w(1:nw:abs(incw))

99999 Format (1X,'Result of scaled vector addition is')
99998 Format (1X,'W =',5F9.4)
    End Program f16ehfe
```

10.2 Program Data

```
F16EHF Example Program Data
   5                                : n
  -1     -1    -1                   : incx, incy and incw
   3.0   -1.0                       : alpha and beta
  -4.0    2.1   3.7    4.5   -6.0    : x
  -3.0   -2.4   6.4   -5.0   -5.1    : y
```

10.3 Program Results

```
 F16EHF Example Program Results

 Result of scaled vector addition is
 W =  -9.0000   8.7000   4.7000  18.5000 -12.9000
```